Partial Derivatives

```
> restart:with(plots):with(DEtools):
```

We again consider the function

$$z = x\, e^{-x^2 - y^2},$$

which we first write as an expression.

```
> z:=x*exp(-x^2-y^2);
```

$$z := x\, e^{-x^2 - y^2}$$

Let's take its partial derivatives. We start with the formal partial Diff with respect to *x*.

```
> Fx:=Diff(z,x);
```

$$Fx := \frac{\partial}{\partial x}\left(x\, e^{-x^2 - y^2}\right)$$

This is the symbolization. Now we get its value.

```
> Fx:=value(%);
```

$$Fx := e^{-x^2 - y^2} - 2\, x^2\, e^{-x^2 - y^2}$$

We'll get right to the value of the partial derivative with respect to *y* by using diff.

```
> Fy:=diff(z,y);
```

$$Fy := -2\, x\, y\, e^{-x^2 - y^2}$$

Now let's do the second partials by using Diff.

```
> Fxx:=Diff(Diff(z,x),x);
```

$$Fxx := \frac{\partial^2}{\partial x^2}\left(x\, e^{-x^2 - y^2}\right)$$

```
> Fxy:=Diff(Diff(z,x),y);
```

$$Fxy := \frac{\partial^2}{\partial y\, \partial x}\left(x\, e^{-x^2 - y^2}\right)$$

```
> Fyx:=Diff(Diff(z,y),x);
```

$$Fyx := \frac{\partial^2}{\partial x\, \partial y}\left(x\, e^{-x^2 - y^2}\right)$$

```
> Fyy:=Diff(Diff(z,y),y);
```

$$Fyy := \frac{\partial^2}{\partial y^2}\left(x\, e^{-x^2 - y^2}\right)$$

Of course, one can evaluate any of these by using value as above. Now let's go straight to the values of the derivatives.

```
> Fxx:=diff(Fx,x);
```

$$Fxx := -6\, x\, e^{-x^2 - y^2} + 4\, x^3\, e^{-x^2 - y^2}$$

```
> Fxy:=diff(Fx,y);
```

$$Fxy := -2\,y\,e^{-x^2-y^2} + 4\,x^2\,y\,e^{-x^2-y^2}$$

```
> Fyx:=diff(Fy,x);
```

$$Fyx := -2\,y\,e^{-x^2-y^2} + 4\,x^2\,y\,e^{-x^2-y^2}$$

Notice that the mixed partials are equal. Just for variety,

```
> Fyy:=diff(diff(z,y),y);
```

$$Fyy := -2\,x\,e^{-x^2-y^2} + 4\,x\,y^2\,e^{-x^2-y^2}$$

Now suppose the function is given as a Maple function.

```
> F:=(x,y)->x*exp(-x^2-y^2);
```

$$F := (x, y) \rightarrow x\,e^{-x^2-y^2}$$

In this situation, we could proceed exactly as above, just replacing each occurance of z with $F(x, y)$. But another option is to use the differential operator D.

```
> Fx:=D[1](F);
```

$$Fx := (x, y) \rightarrow e^{-x^2-y^2} - 2\,x^2\,e^{-x^2-y^2}$$

```
> Fy:=D[2](F);
```

$$Fy := (x, y) \rightarrow -2\,x\,y\,e^{-x^2-y^2}$$

```
> Fxx:=D[1,1](F);
```

$$Fxx := (x, y) \rightarrow -6\,x\,e^{-x^2-y^2} + 4\,x^3\,e^{-x^2-y^2}$$

```
> Fxy:=D[1,2](F);
```

$$Fxy := (x, y) \rightarrow -2\,y\,e^{-x^2-y^2} + 4\,x^2\,y\,e^{-x^2-y^2}$$

```
> Fxy:=D[2](D[1](F));
```

$$Fxy := (x, y) \rightarrow -2\,y\,e^{-x^2-y^2} + 4\,x^2\,y\,e^{-x^2-y^2}$$

```
> Fyx:=D[2,1](F);
```

$$Fyx := (x, y) \rightarrow -2\,y\,e^{-x^2-y^2} + 4\,x^2\,y\,e^{-x^2-y^2}$$

```
> Fyy:=D[2,2](F);
```

$$Fyy := (x, y) \rightarrow -2\,x\,e^{-x^2-y^2} + 4\,x\,y^2\,e^{-x^2-y^2}$$